
NAG Library Routine Document

D02GAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

D02GAF solves a two-point boundary value problem with assigned boundary values for a system of
ordinary differential equations, using a deferred correction technique and a Newton iteration.

2 Specification

SUBROUTINE D02GAF (U, V, N, A, B, TOL, FCN, MNP, X, Y, NP, W, LW, IW,
LIW, IFAIL)

&

INTEGER N, MNP, NP, LW, IW(LIW), LIW, IFAIL
REAL (KIND=nag_wp) U(N,2), V(N,2), A, B, TOL, X(MNP), Y(N,MNP), W(LW)
EXTERNAL FCN

3 Description

D02GAF solves a two-point boundary value problem for a system of n differential equations in the
interval [a; b]. The system is written in the form:

y0i ¼ fi x; y1; y2; . . . ; ynð Þ; i ¼ 1; 2; . . . ; n ð1Þ
and the derivatives fi are evaluated by FCN. Initially, n boundary values of the variables yi must be
specified, some at a and some at b. You must supply estimates of the remaining n boundary values and
all the boundary values are used in constructing an initial approximation to the solution. This
approximate solution is corrected by a finite difference technique with deferred correction allied with a
Newton iteration to solve the finite difference equations. The technique used is described fully in

Pereyra (1979). The Newton iteration requires a Jacobian matrix
@fi
@yj

and this is calculated by numerical

differentiation using an algorithm described in Curtis et al. (1974).

You supply an absolute error tolerance and may also supply an initial mesh for the construction of the
finite difference equations (alternatively a default mesh is used). The algorithm constructs a solution on
a mesh defined by adding points to the initial mesh. This solution is chosen so that the error is
everywhere less than your tolerance and so that the error is approximately equidistributed on the final
mesh. The solution is returned on this final mesh.

If the solution is required at a few specific points then these should be included in the initial mesh. If on
the other hand the solution is required at several specific points then you should use the interpolation
routines provided in Chapter E01 if these points do not themselves form a convenient mesh.

4 References

Curtis A R, Powell M J D and Reid J K (1974) On the estimation of sparse Jacobian matrices J. Inst.
Maths. Applics. 13 117–119

Pereyra V (1979) PASVA3: An adaptive finite-difference Fortran program for first order nonlinear,
ordinary boundary problems Codes for Boundary Value Problems in Ordinary Differential Equations.
Lecture Notes in Computer Science (eds B Childs, M Scott, J W Daniel, E Denman and P Nelson) 76
Springer–Verlag

5 Arguments

1: UðN; 2Þ – REAL (KIND=nag_wp) array Input

On entry: Uði; 1Þ must be set to the known or estimated value of yi at a and Uði; 2Þ must be set to
the known or estimated value of yi at b, for i ¼ 1; 2; . . . ;n.

2: VðN; 2Þ – REAL (KIND=nag_wp) array Input

On entry: Vði; jÞ must be set to 0:0 if Uði; jÞ is a known value and to 1:0 if Uði; jÞ is an estimated
value, for i ¼ 1; 2; . . . ;n and j ¼ 1; 2.

Constraint: precisely n of the Vði; jÞ must be set to 0:0, i.e., precisely n of the Uði; jÞ must be
known values, and these must not be all at a or all at b.

3: N – INTEGER Input

On entry: n, the number of equations.

Constraint: N � 2.

4: A – REAL (KIND=nag_wp) Input

On entry: a, the left-hand boundary point.

5: B – REAL (KIND=nag_wp) Input

On entry: b, the right-hand boundary point.

Constraint: B > A.

6: TOL – REAL (KIND=nag_wp) Input

On entry: a positive absolute error tolerance. If

a ¼ x1 < x2 < � � � < xNP ¼ b

is the final mesh, zj xið Þ is the jth component of the approximate solution at xi, and yj xð Þ is the
jth component of the true solution of equation (1) (see Section 3) and the boundary conditions,
then, except in extreme cases, it is expected that

zj xið Þ � yj xið Þ�� �� � TOL; i ¼ 1; 2; . . . ;NP and j ¼ 1; 2; . . . ;n: ð2Þ
Constraint: TOL > 0:0.

7: FCN – SUBROUTINE, supplied by the user. External Procedure

FCN must evaluate the functions fi (i.e., the derivatives y0i), for i ¼ 1; 2; . . . ; n, at a general point
x.

The specification of FCN is:

SUBROUTINE FCN (X, Y, F)

REAL (KIND=nag_wp) X, Y(*), F(*)

In the description of the arguments of D02GAF below, n denotes the actual value of N in the
call of D02GAF.

1: X – REAL (KIND=nag_wp) Input

On entry: x, the value of the argument.

2: Yð�Þ – REAL (KIND=nag_wp) array Input

On entry: yi, for i ¼ 1; 2; . . . ; n, the value of the argument.

3: Fð�Þ – REAL (KIND=nag_wp) array Output

On exit: the values of fi, for i ¼ 1; 2; . . . ; n.

FCN must either be a module subprogram USEd by, or declared as EXTERNAL in, the (sub)
program from which D02GAF is called. Arguments denoted as Input must not be changed by
this procedure.

8: MNP – INTEGER Input

On entry: the maximum permitted number of mesh points.

Constraint: MNP � 32.

9: XðMNPÞ – REAL (KIND=nag_wp) array Input/Output

On entry: if NP � 4 (see NP), the first NP elements must define an initial mesh. Otherwise the
elements of X need not be set.

Constraint:

A ¼ Xð1Þ < Xð2Þ < � � � < XðNPÞ ¼ B; NP � 4: ð3Þ
On exit: Xð1Þ;Xð2Þ; . . . ;XðNPÞ define the final mesh (with the returned value of NP) satisfying
the relation (3).

10: YðN;MNPÞ – REAL (KIND=nag_wp) array Output

On exit: the approximate solution zj xið Þ satisfying (2), on the final mesh, that is

Yðj; iÞ ¼ zj xið Þ; i ¼ 1; 2; . . . ;NP and j ¼ 1; 2; . . . ; n;

where NP is the number of points in the final mesh.

The remaining columns of Y are not used.

11: NP – INTEGER Input/Output

On entry: determines whether a default or user-supplied mesh is used.

NP ¼ 0
A default value of 4 for NP and a corresponding equispaced mesh Xð1Þ;Xð2Þ; . . . ;XðNPÞ
are used.

NP � 4
You must define an initial mesh using the array X as described.

Constraint: NP ¼ 0 or 4 � NP � MNP.

On exit: the number of points in the final (returned) mesh.

12: WðLWÞ – REAL (KIND=nag_wp) array Workspace
13: LW – INTEGER Input

On entry: the dimension of the array W as declared in the (sub)program from which D02GAF is
called.

Constraint: LW � MNP� 3N2 þ 6Nþ 2
� �þ 4N2 þ 4N.

14: IWðLIWÞ – INTEGER array Workspace
15: LIW – INTEGER Input

On entry: the dimension of the array IW as declared in the (sub)program from which D02GAF is
called.

Constraint: LIW � MNP� 2Nþ 1ð Þ þ N2 þ 4Nþ 2.

16: IFAIL – INTEGER Input/Output

For this routine, the normal use of IFAIL is extended to control the printing of error and warning
messages as well as specifying hard or soft failure (see Section 3.4 in How to Use the NAG
Library and its Documentation).

On entry: IFAIL must be set to a value with the decimal expansion cba, where each of the
decimal digits c, b and a must have a value of 0 or 1.

a ¼ 0 specifies hard failure, otherwise soft failure;

b ¼ 0 suppresses error messages, otherwise error messages will be printed (see Section 6);

c ¼ 0 suppresses warning messages, otherwise warning messages will be printed (see Section 6).

The recommended value for inexperienced users is 110 (i.e., hard failure with all messages
printed).

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

One or more of the arguments N, TOL, NP, MNP, LW or LIW has been incorrectly set, or B � A,
or the condition (3) on X is not satisfied, or the number of known boundary values (specified by
V) is not N.

IFAIL ¼ 2

The Newton iteration has failed to converge. This could be due to there being too few points in
the initial mesh or to the initial approximate solution being too inaccurate. If this latter reason is
suspected you should use D02RAF instead. If the warning ‘Jacobian matrix is singular’ is printed
this could be due to specifying zero estimated boundary values and these should be varied. This
warning could also be printed in the unlikely event of the Jacobian matrix being calculated
inaccurately. If you cannot make changes to prevent the warning then D02RAF should be used.

IFAIL ¼ 3

The Newton iteration has reached round-off level. It could be, however, that the answer returned
is satisfactory. This error might occur if too much accuracy is requested.

IFAIL ¼ 4

A finer mesh is required for the accuracy requested; that is MNP is not large enough.

IFAIL ¼ 5

A serious error has occurred in a call to D02GAF. Check all array subscripts and subroutine
argument lists in calls to D02GAF. Seek expert help.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The solution returned by the routine will be accurate to your tolerance as defined by the relation (2)
except in extreme circumstances. If too many points are specified in the initial mesh, the solution may
be more accurate than requested and the error may not be approximately equidistributed.

8 Parallelism and Performance

D02GAF is not thread safe and should not be called from a multithreaded user program. Please see
Section 3.12.1 in How to Use the NAG Library and its Documentation for more information on thread
safety.

D02GAF is not threaded in any implementation.

9 Further Comments

The time taken by D02GAF depends on the difficulty of the problem, the number of mesh points (and
meshes) used, the number of Newton iterations and the number of deferred corrections.

You are strongly recommended to set IFAIL to obtain self-explanatory error messages, and also
monitoring information about the course of the computation. You may select the unit numbers on which
this output is to appear by calls of X04AAF (for error messages) or X04ABF (for monitoring
information) – see Section 10 for an example. Otherwise the default unit numbers will be used, as
specified in the Users' Note.

A common cause of convergence problems in the Newton iteration is that you have specified too few
points in the initial mesh. Although the routine adds points to the mesh to improve accuracy it is unable
to do so until the solution on the initial mesh has been calculated in the Newton iteration.

If you specify zero known and estimated boundary values, the routine constructs a zero initial
approximation and in many cases the Jacobian is singular when evaluated for this approximation,
leading to the breakdown of the Newton iteration.

You may be unable to provide a sufficiently good choice of initial mesh and estimated boundary values,
and hence the Newton iteration may never converge. In this case the continuation facility provided in
D02RAF is recommended.

In the case where you wish to solve a sequence of similar problems, the final mesh from solving one
case is strongly recommended as the initial mesh for the next.

10 Example

This example solves the differential equation

y000 ¼ �yy00 � � 1� y02
� �

with boundary conditions

y 0ð Þ ¼ y0 0ð Þ ¼ 0; y0 10ð Þ ¼ 1

for � ¼ 0:0 and � ¼ 0:2 to an accuracy specified by TOL ¼ 1:0E�3. We solve first the simpler problem

with � ¼ 0:0 using an equispaced mesh of 26 points and then we solve the problem with � ¼ 0:2 using
the final mesh from the first problem.

Note the call to X04ABF prior to the call to D02GAF.

10.1 Program Text

! D02GAF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module d02gafe_mod

! Data for D02GAF example program

! .. Use Statements ..
Use nag_library, Only: nag_wp

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private
Public :: fcn

! .. Parameters ..
Real (Kind=nag_wp), Parameter, Public :: one = 1.0_nag_wp
Real (Kind=nag_wp), Parameter, Public :: zero = 0.0_nag_wp
Integer, Parameter, Public :: iset = 1, n = 4, nin = 5, nout = 6

Contains
Subroutine fcn(x,y,f)

! .. Scalar Arguments ..
Real (Kind=nag_wp), Intent (In) :: x

! .. Array Arguments ..
Real (Kind=nag_wp), Intent (Out) :: f(*)
Real (Kind=nag_wp), Intent (In) :: y(*)

! .. Executable Statements ..
f(1) = y(2)
f(2) = y(3)
f(3) = -y(1)*y(3) - y(4)*(1.0E0_nag_wp-y(2)*y(2))
f(4) = 0.0_nag_wp
Return

End Subroutine fcn
End Module d02gafe_mod
Program d02gafe

! D02GAF Example Main Program

! .. Use Statements ..
Use nag_library, Only: d02gaf, nag_wp, x04abf
Use d02gafe_mod, Only: fcn, iset, n, nin, nout, one, zero

! .. Implicit None Statement ..
Implicit None

! .. Local Scalars ..
Real (Kind=nag_wp) :: a, b, beta, h, tol
Integer :: i, ifail, j, k, liw, lw, mnp, np, &

outchn
! .. Local Arrays ..

Real (Kind=nag_wp), Allocatable :: u(:,:), v(:,:), w(:), x(:), y(:,:)
Integer, Allocatable :: iw(:)

! .. Intrinsic Procedures ..
Intrinsic :: real

! .. Executable Statements ..
Write (nout,*) ’D02GAF Example Program Results’

! Skip heading in data file
Read (nin,*)

! n: number of differential equations
! mnp: maximum permitted number of mesh points.

Read (nin,*) mnp
liw = mnp*(2*n+1) + n*n + 4*n + 2
lw = mnp*(3*n*n+6*n+2) + 4*n*n + 4*n
Allocate (iw(liw),u(n,2),v(n,2),w(lw),x(mnp),y(n,mnp))

! tol: positive absolute error tolerance

! np : determines whether a default or user-supplied mesh is used.
! a : left-hand boundary point, b: right-hand boundary point.

Read (nin,*) tol
Read (nin,*) np
Read (nin,*) a, b
outchn = nout
Call x04abf(iset,outchn)
beta = zero
u(1:n,1:2) = zero
v(1:n,1:2) = zero
v(1,2) = one
v(3,1) = one
v(3,2) = one
v(4,2) = one
u(2,2) = one
u(1,2) = b
x(1) = a
h = (b-a)/real(np-1,kind=nag_wp)
Do i = 2, np - 1

x(i) = x(i-1) + h
End Do
x(np) = b
beta = zero

loop: Do k = 1, 2
u(4,1) = beta
u(4,2) = beta

! ifail: behaviour on error exit
! =1 for quiet-soft exit
! * Set ifail to 111 to obtain monitoring information *

ifail = 1
Call d02gaf(u,v,n,a,b,tol,fcn,mnp,x,y,np,w,lw,iw,liw,ifail)

If (ifail>=0) Then
Write (nout,99999) ’Problem with BETA = ’, beta

End If
If (ifail==0 .Or. ifail==3) Then

Write (nout,*)
If (ifail==3) Then

Write (nout,*) ’ IFAIL = 3’
End If
Write (nout,99998) np
Write (nout,99997)
Write (nout,99996)(x(i),(y(j,i),j=1,3),i=1,np)
beta = beta + 0.2E0_nag_wp

Else
Write (nout,99995) ifail
Exit loop

End If
End Do loop

99999 Format (/,1X,A,F7.2)
99998 Format (1X,’Solution on final mesh of ’,I2,’ points’)
99997 Format (1X,’ X(I) Y1(I) Y2(I) Y3(I)’)
99996 Format (1X,F11.3,3F13.4)
99995 Format (1X,/,1X,’ ** D02GAF returned with IFAIL = ’,I5)

End Program d02gafe

10.2 Program Data

D02GAF Example Program Data
40 : mnp
1.0E-3 : tol
26 : np
0.0 10.0 : a, b

10.3 Program Results

D02GAF Example Program Results

Problem with BETA = 0.00

Solution on final mesh of 26 points
X(I) Y1(I) Y2(I) Y3(I)
0.000 0.0000 0.0000 0.4695
0.400 0.0375 0.1876 0.4673
0.800 0.1497 0.3719 0.4511
1.200 0.3336 0.5450 0.4104
1.600 0.5828 0.6963 0.3424
2.000 0.8864 0.8163 0.2558
2.400 1.2309 0.9009 0.1678
2.800 1.6026 0.9529 0.0953
3.200 1.9900 0.9805 0.0464
3.600 2.3851 0.9930 0.0193
4.000 2.7834 0.9978 0.0069
4.400 3.1829 0.9994 0.0021
4.800 3.5828 0.9999 0.0006
5.200 3.9828 1.0000 0.0001
5.600 4.3828 1.0000 0.0000
6.000 4.7828 1.0000 0.0000
6.400 5.1828 1.0000 0.0000
6.800 5.5828 1.0000 0.0000
7.200 5.9828 1.0000 -0.0000
7.600 6.3828 1.0000 0.0000
8.000 6.7828 1.0000 -0.0000
8.400 7.1828 1.0000 0.0000
8.800 7.5828 1.0000 -0.0000
9.200 7.9828 1.0000 0.0000
9.600 8.3828 1.0000 -0.0000

10.000 8.7828 1.0000 -0.0000

Problem with BETA = 0.20

Solution on final mesh of 26 points
X(I) Y1(I) Y2(I) Y3(I)
0.000 0.0000 0.0000 0.6865
0.400 0.0528 0.2584 0.6040
0.800 0.2020 0.4814 0.5091
1.200 0.4324 0.6636 0.4001
1.600 0.7268 0.8007 0.2860
2.000 1.0670 0.8939 0.1821
2.400 1.4368 0.9498 0.1017
2.800 1.8233 0.9791 0.0492
3.200 2.2180 0.9924 0.0206
3.600 2.6162 0.9976 0.0074
4.000 3.0157 0.9993 0.0023
4.400 3.4156 0.9998 0.0006
4.800 3.8155 1.0000 0.0001
5.200 4.2155 1.0000 0.0000
5.600 4.6155 1.0000 0.0000
6.000 5.0155 1.0000 0.0000
6.400 5.4155 1.0000 -0.0000
6.800 5.8155 1.0000 -0.0000
7.200 6.2155 1.0000 -0.0000
7.600 6.6155 1.0000 -0.0000
8.000 7.0155 1.0000 -0.0000
8.400 7.4155 1.0000 -0.0000
8.800 7.8155 1.0000 -0.0000
9.200 8.2155 1.0000 -0.0000
9.600 8.6155 1.0000 -0.0000

10.000 9.0155 1.0000 -0.0000

